
Introduction

This document summarizes the two proposed changes to the ABNF productions currently in XRI Syntax 2.0 Committee Draft 01 (http://www.oasis-open.org/committees/download.php/11852/xri-syntax-v2.0-cd-01.pdf). 

Note: These proposed changes are independent of each other. The first proposes to simplify and better align XRI ABNF with RFC 3986 URI ABNF. This proposal has priority for advancing XRI Syntax to an OASIS Standard. The second proposes the addition of productions that align with HTTP proxy resolution in the Resolution specification. This does not have priority for advancing XRI Syntax to an OASIS Standard, however if consensus on the preferred format of HXRIs can be reached quickly, including in XRI 2.0 will not necessitate revising XRI Syntax when XRI Resolution and XRI Metadata are ready to advance to an OASIS Standard.
Proposal #1: Support Empty Path Segments

Motivation

The OASIS XDI TC recently reached consensus to use double forward slash (“//”) in XRIs as a special segment delimiter in XDI addressing. Subsequent analysis revealed that while the ABNF of RFCs 3986 and 3987 allows // anywhere in the path portion of an absolute URI/IRI, including at the start of the path following the authority segment, the ABNF of XRI 2.0 CD 01 does not allow // at the start of the path following the authority segment of an absolute XRI.

The motivation for this proposal is to fix this discrepancy, as well as to slightly simplify XRI 2.0 syntax.

Requirements
· Allow XRI path segments at any level of the path to be empty in order to align XRI and IRI/URI path syntax.
Analysis/Design
While the 3986/3987 ABNF allows multiple consecutive forward slashes anywhere in the path portion of an absolute URI/IRI, it does not allow more than a single forward slash at the start of a relative URI/IRI reference, i.e., a path that is not preceded by an authority segment or scheme name. The reason for this is that a URI/IRI can be relative to a scheme. For example, the relative URI reference “//foo/bar”can be used with both the http and ftp schemes:


http://foo/bar

ftp://foo/bar
As abstract identifiers, XRIs are not scheme-relative. Absolute XRIs must either begin with “xri://”, a GCS character, a cross-reference, or the i-authority production 
([ iuserinfo "@" ] ihost [ ":" port ]). Thus a relative XRI reference that begins with a forward slash character (single, double, or any other multiple) can always be interpreted as a path and never as an authority segment.

This means XRI 2.0 CD 02 ABNF can use a simpler path syntax that resolves the discrepancy with absolute URI/IRI syntax with regard to multiple forward slashes in the path portion.

Proposed ABNF

This section shows the proposed revised ABNF that fulfills these requirements (see Appendix A for the current Committee Draft 01 ABNF). A summary of the changes:
· Three productions were removed because they are no longer needed: xri-path-absolute, xri-segment-nz, and xri-subseg-od-nz.
· The four lines that were changed are xri-hier-part, relative-XRI-ref, xri-path, and xri-path-noscheme (note that the fourth change, to the xri-path-noscheme production, was not a change in substance but only in the name used to refer to the new simplified xri-path production).
Each of these changes is shown in bold in the complete ABNF below (not including the IRI productions – see Appendix A for those): 
XRI               = [ "xri://" ] xri-hier-part [ "?" iquery ]
                  [ "#" ifragment ]  

xri-hier-part     = ( xri-authority / iauthority ) [ xri-path ]
XRI-reference     = XRI
                  / relative-XRI-ref  

absolute-XRI      = [ "xri://" ] xri-hier-part [ "?" iquery ]

xri-value
= xri-no-scheme / relative-XRI-ref
xri-no-scheme
= xri-hier-part [ "?" iquery ] [ "#" ifragment ]  

relative-XRI-ref  = xri-path-noscheme [ "?" iquery ] [ "#" ifragment ]

xri-authority     = gcs-authority
                  / xref-authority

gcs-authority     = pgcs-authority 


/ rgcs-authority

pgcs-authority    = "!" xri-subseg-pt-nz *xri-subseg
rgcs-authority    = rgcs-char xri-segment

rgcs-char         = "=" / "@" / "+" / "$" 

xref-authority    = xref *xri-subseg

xref              = "(" ( XRI-reference / IRI ) ")"

xri-path          = *( "/" xri-segment )
xri-path-noscheme = xri-subseg-od-nx *xri-subseg-nc xri-path
xri-segment       = xri-subseg-od *xri-subseg

xri-subseg        = ( "*" / "!" ) (xref / *xri-pchar)

xri-subseg-nc     = ( "*" / "!" ) (xref / *xri-pchar-nc)

xri-subseg-od     = [ "*" / "!" ] (xref / *xri-pchar)

xri-subseg-od-nx  = [ "*" / "!" ] 1*xri-pchar-nc

xri-subseg-pt-nz  = "!" (xref / 1*xri-pchar)
xri-pchar         = iunreserved / pct-encoded / xri-sub-delims / ":"

xri-pchar-nc      = iunreserved / pct-encoded / xri-sub-delims

xri-reserved      = xri-gen-delims / xri-sub-delims

xri-gen-delims
= ":" / "/" / "?" / "#" / "[" / "]" / "(" / ")"
                / "*" / "!" / rgcs-char

xri-sub-delims    = "&" / ";" / "," / "'"

Proposal #2: Add HXRI and QXRI Productions
Motivation

The motivation for these productions is explained in the Local and Proxy Resolution Proposal (http://www.oasis-open.org/committees/download.php/14296/local-and-proxy-resolution-proposal-v1.doc). In summary:

· The HXRI (HTTP XRI) production defines a standard way in which any XRI can be expressed as an HTTP(S) URI and thus become immediately resolvable using HTTP proxy resolution, which has enormous advantages in terms of backwards compatability with existing HTTP(S) infrastructure.

· The QXRI (Query XRI) production defines the constraints on the XRI embedded in an HXRI.

Requirements

The key requirements are:

· To have a single production that defines the ABNF for an XRI in HTTP(S) format.

· To minimize the processing necessary by user agents, search engines, and other markup processors to recognize that an HTTP(S) URI is an HXRI.

· For the parsing test to determine if an HTTP URI is an HXRI to have a minimal chance of producing false positives.

· To establish at least one HXRI format that is easy for humans to remember and type.

· For a QXRI not to be ambiguous.

Analysis/Design

The requirements contain some natural conflicts, e.g., the easier it is for a human to remember and type an HXRI, the harder it is for it to be unambiguous to a markup processor.

The proposed design strikes a middle ground by allowing QXRIs only as the entire path or entire query component of an HTTP(S) URI. In both cases the potential for ambiguity of the QXRI value is minimized by restricting it to either:
· A fully-qualified XRI (FQXRI) that must begin with “xri://”, or

· A xri-authority-qualified XRI (XQXRI) that must begin with an XRI authority character—either a GCS character (=, @, +, $, !) or a left parenthese for a cross-reference.

Proposed ABNF

HXRI
= ( "http://" / "https://" ) iauthority 


( "/" / "?" ) QXRI

QXRI
= FQXRI / XQXRI

FQXRI
= "xri://" xri-hier-part [ "?" iquery ]
                  [ "#" ifragment ]
XQXRI
= xri-authority [ xri-path ] [ "?" iquery ]
                  [ "#" ifragment ]

Appendix A: Current XRI 2.0 CD 01 ABNF
Note that productions that were revised in the first proposal above are shown in italics; those that were dropped are shown in bold.

XRI               = [ "xri://" ] xri-hier-part [ "?" iquery ]
                  [ "#" ifragment ]  

xri-hier-part     = ( xri-authority / iauthority ) [ xri-path-absolute ]
                  / ipath-empty  

XRI-reference     = XRI
                  / relative-XRI-ref  

absolute-XRI      = [ "xri://" ] xri-hier-part [ "?" iquery ]

xri-value
= xri-no-scheme / relative-XRI-ref
xri-no-scheme
= xri-hier-part [ "?" iquery ]
                  [ "#" ifragment ]  

relative-XRI-ref  = xri-path [ "?" iquery ] [ "#" ifragment ]

xri-authority     = gcs-authority
                  / xref-authority

gcs-authority     = pgcs-authority / rgcs-authority

pgcs-authority    = "!" xri-subseg-pt-nz *xri-subseg
rgcs-authority    = rgcs-char xri-segment

rgcs-char         = "=" / "@" / "+" / "$" 

xref-authority    = xref *xri-subseg

xref              = "(" ( XRI-reference / IRI ) ")"

xri-path          = xri-path-absolute
                  / xri-path-noscheme
                  / ipath-empty  

xri-path-absolute = "/" [ xri-segment-nz *( "/" xri-segment ) ]

xri-path-noscheme = xri-subseg-od-nx *xri-subseg-nc *( "/" xri-segment )

xri-segment       = xri-subseg-od *xri-subseg

xri-segment-nz    = xri-subseg-od-nz *xri-subseg

xri-subseg        = ( "*" / "!" ) (xref / *xri-pchar)

xri-subseg-nc     = ( "*" / "!" ) (xref / *xri-pchar-nc)

xri-subseg-od     = [ "*" / "!" ] (xref / *xri-pchar)

xri-subseg-od-nz  = [ "*" / "!" ] (xref / 1*xri-pchar)

xri-subseg-od-nx  = [ "*" / "!" ] 1*xri-pchar-nc

xri-subseg-pt-nz  = "!" (xref / 1*xri-pchar)
xri-pchar         = iunreserved / pct-encoded / xri-sub-delims / ":"

xri-pchar-nc      = iunreserved / pct-encoded / xri-sub-delims

xri-reserved      = xri-gen-delims / xri-sub-delims

xri-gen-delims
= ":" / "/" / "?" / "#" / "[" / "]" / "(" / ")"
                / "*" / "!" / rgcs-char
xri-sub-delims    = "&" / ";" / "," / "'"

IRI               = scheme ":" ihier-part [ "?" iquery ]
                  [ "#" ifragment ]

scheme            = ALPHA *( ALPHA / DIGIT / "+" / "-" / "." )

ihier-part        = "//" iauthority ipath-abempty
                  / ipath-abs
                  / ipath-rootless
                  / ipath-empty

iauthority        = [ iuserinfo "@" ] ihost [ ":" port ]

iuserinfo         = *( iunreserved / pct-encoded / sub-delims / ":" )

ihost             = IP-literal / IPv4address / ireg-name

IP-literal        = "[" ( IPv6address / IPvFuture  ) "]"

IPvFuture         = "v" 1*HEXDIG "." 1*( unreserved / sub-delims / ":" )

IPv6address       =                            6( h16 ":" ) ls32
                  /                       "::" 5( h16 ":" ) ls32
                  / [               h16 ] "::" 4( h16 ":" ) ls32
                  / [ *1( h16 ":" ) h16 ] "::" 3( h16 ":" ) ls32
                  / [ *2( h16 ":" ) h16 ] "::" 2( h16 ":" ) ls32
                  / [ *3( h16 ":" ) h16 ] "::"    h16 ":"   ls32
                  / [ *4( h16 ":" ) h16 ] "::"              ls32
                  / [ *5( h16 ":" ) h16 ] "::"              h16
                  / [ *6( h16 ":" ) h16 ] "::"

ls32              = ( h16 ":" h16 ) / IPv4address

h16               = 1*4HEXDIG

IPv4address       = dec-octet "." dec-octet "." dec-octet "." dec-octet

dec-octet         = DIGIT                ; 0-9
                  / %x31-39 DIGIT        ; 10-99
                  / "1" 2DIGIT           ; 100-199
                  / "2" %x30-34 DIGIT    ; 200-249
                  / "25" %x30-35         ; 250-255

ireg-name         = *( iunreserved / pct-encoded / sub-delims )

port              = *DIGIT

ipath-abempty     = *( "/" isegment )

ipath-abs         = "/" [ isegment-nz *( "/" isegment ) ]

ipath-rootless    = isegment-nz *( "/" isegment )

ipath-empty       = 0<ipchar>

isegment          = *ipchar

isegment-nz       = 1*ipchar

iquery            = *( ipchar / iprivate / "/" / "?" )

iprivate          = %xE000-F8FF / %xF0000-FFFFD / %x100000-10FFFD

ifragment         = *( ipchar / "/" / "?" )

ipchar            = iunreserved / pct-encoded / sub-delims / ":" / "@"
iquery            = *( ipchar / iprivate / "/" / "?" )

ifragment         = *( ipchar / "/" / "?" )

iunreserved       = ALPHA / DIGIT / "-" / "." / "_" / "~" / ucschar

pct-encoded       = "%" HEXDIG HEXDIG

ucschar           = %xA0-D7FF / %xF900-FDCF / %xFDF0-FFEF
                  / %x10000-1FFFD / %x20000-2FFFD / %x30000-3FFFD
                  / %x40000-4FFFD / %x50000-5FFFD / %x60000-6FFFD
                  / %x70000-7FFFD / %x80000-8FFFD / %x90000-9FFFD
                  / %xA0000-AFFFD / %xB0000-BFFFD / %xC0000-CFFFD
                  / %xD0000-DFFFD / %xE1000-EFFFD

reserved          = gen-delims / sub-delims

gen-delims        = ":" / "/" / "?" / "#" / "[" / "]" / "@"

sub-delims        = "!" / "$" / "&" / "'" / "(" / ")"
                  / "*" / "+" / "," / ";" / "="

unreserved        = ALPHA / DIGIT / "-" / "." / "_" / "~"


